Asymptotic of sequence A227403

(*Václav Kotěšovec, published 21.9.2013*)

In the OEIS (On-Line Encyclopedia of Integer Sequences) published Paul D. Hanna 20.9.2003 a sequence A227403

$$a_n = \sum_{k=0}^{n} \binom{n^2}{nk} \binom{nk}{k^2}$$

I found following limit

$$\lim_{n\to\infty} \left(\sum_{k=0}^{n} \binom{n^2}{nk} \binom{nk}{k^2} \right)^{\frac{1}{n^2}} = r^{-\frac{(1+r)^2}{2r}} = 2.93544172048274\ldots$$

where $r = 0.6032326837741362\ldots$ is the root of the equation

$$(1-r)^{2r} = r^{2r+1}$$

Proof: We find the maximal term with the help of Stirling's approximation, derivative must be zero

```
stirling[n_] := n^n / E^n * Sqrt[2 * Pi * n];
binom[n_, k_] := stirling[n] / stirling[k] / stirling[n - k];

FullSimplify[D[binom[n^2, n * n * r] * binom[n * n * r, (n * r)^2], r]]
```

$$-\frac{1}{4\pi} \left(n^2\right)^{\frac{3}{2}+n^2} \left(-n^2 (-1+r)\right)^{-\frac{1}{2}+n^2 (-1+r)} \left(-n^2 (-1+r)\, r\right)^{-\frac{3}{2}+n^2 (-1+r)\, r} \left(n^2 r^2\right)^{-\frac{1}{2}-n^2 r^2}$$
$$\left(3 - 5 r + 2 n^2 (-1+r)\, r \left(\text{Log}\left[-n^2 (-1+r)\right] + (-1+2 r)\, \text{Log}\left[-n^2 (-1+r)\, r\right] - 2 r\, \text{Log}\left[n^2 r^2\right]\right)\right)$$

```
Limit[
 (3 - 5 r + 2 n^2 (-1 + r) r (Log[-n^2 (-1 + r)] + (-1 + 2 r) Log[-n^2 (-1 + r) r] - 2 r Log[n^2 r^2])) /
  n^2, n → Infinity]
```

$$2 (-1+r)\, r \left(\text{Log}[1-r] + (-1+2 r)\, \text{Log}[-(-1+r)\, r] - 2 r\, \text{Log}\left[r^2\right]\right)$$

The maximal term is asymptotically at position $k = r * n$, where r is the root of the equation

$$(1-r)^{2r} = r^{2r+1}$$

Numerically

```
FindRoot[(1 - r)^(2 * r) == r^(2 * r + 1), {r, 1/2}, WorkingPrecision → 50]
{r → 0.60323268377413620622019265094866822042096251421175}
```

Following graph is in the logarithmical scale

```
n = 1000;
ListPlot[Table[Log[Binomial[n^2, n * k] * Binomial[n * k, k^2]], {k, 0, n}]]
```



Value in the maximum satisfy the inequality

$$\binom{n^2}{rn^2}\binom{rn^2}{r^2n^2} \le \sum_{k=0}^{n}\binom{n^2}{nk}\binom{nk}{k^2} \le n * \binom{n^2}{rn^2}\binom{rn^2}{r^2n^2}$$

and

$$\lim_{n\to\infty}\left(\sum_{k=0}^{n}\binom{n^2}{nk}\binom{nk}{k^2}\right)^{\frac{1}{n^2}} = \lim_{n\to\infty}\left(\binom{n^2}{rn^2}\binom{rn^2}{r^2n^2}\right)^{\frac{1}{n^2}}$$

```
FullSimplify[PowerExpand[(binom[n^2, n*n*r]*binom[n*n*r, (n*r)^2])^(1/n^2)]]
```

$$n^{2-2r^2}(2\pi)^{-\frac{1}{n^2}}\left(-n^2(-1+r)\right)^{-1-\frac{1}{2n^2}+r} r^{-\frac{1}{n^2}-2r^2}\left(-n^2(-1+r)r\right)^{-\frac{1}{2n^2}+(-1+r)r}$$

This expression can be simplified after substitution

$$1-r = r^{\frac{2r+1}{2r}}$$

```
FullSimplify[
 PowerExpand[n^(2-2 r^2) (2 π)^(-1/n^2) (n^2*(r^((2*r+1)/(2*r))))^(-1-1/(2 n^2)+r) r^(-1/n^2-2 r^2)
  (n^2*(r^((2*r+1)/(2*r)))*r)^(-1/(2 n^2)+(-1+r) r)]]
```

$$n^{-\frac{2}{n^2}}(2\pi)^{-\frac{1}{n^2}} r^{-\frac{1+5r+n^2(1+r)^2}{2n^2 r}}$$

```
Limit[n^(-2/n^2) (2 π)^(-1/n^2) r^(-(1+5 r+n^2 (1+r)^2)/(2 n^2 r)), n → Infinity]
```

$$r^{-\frac{(1+r)^2}{2r}}$$

and final result is

$$\lim_{n\to\infty}\left(\sum_{k=0}^{n}\binom{n^2}{nk}\binom{nk}{k^2}\right)^{\frac{1}{n^2}} = \lim_{n\to\infty}\left(\binom{n^2}{rn^2}\binom{rn^2}{r^2n^2}\right)^{\frac{1}{n^2}} = r^{-\frac{(1+r)^2}{2r}} = 2.93544172048274\ldots$$

Numerical verification:

```
ListPlot[Table[(Sum[Binomial[n^2, n*k]*Binomial[n*k, k^2], {k, 0, n}])^(1/n^2),
  {n, 1, 200}]]
```



This was a rigorous proof and we obtain the main asymptotic term. Following results are experimental. Let

$$c = r^{-\frac{(1+r)^2}{2r}} = 2.93544172048274 \ldots$$

Following limit does not exist,

$$\lim_{n\to\infty} \frac{a_n}{c^{n^2}} * n^2$$

but graph of this function is interesting:

```
ListPlot[
 Table[(Sum[Binomial[n^2, n*k]*Binomial[n*k, k^2], {k, 0, n}]) /
   c^(n^2)*n^2, {n, 1, 200}]]
```



Or, with more values

```
c = r^(-(1 + r)^2/(2*r)) /. FindRoot[(1 - r)^(2*r) == r^(2*r + 1), {r, 1/2},
   WorkingPrecision → 100];
val = ParallelTable[(Sum[Binomial[n^2, n*k]*Binomial[n*k, k^2], {k, 0, n}]) / c^(n^2) *
   n^2, {n, 1, 1000}];
ListPlot[val]
```



My **conjecture** is following:

$$a_n = \sum_{k=0}^{n} \binom{n^2}{nk}\binom{nk}{k^2} \sim \frac{c^{n^2}}{n^2} * \left(g + h * cos\left(2\pi\left(\frac{n}{p} + \frac{d}{5}\right)\right)\right)$$

where

$$d = mod(n,5) = n - 5\left\lfloor\frac{n}{5}\right\rfloor$$

period

$$p = 309.3 \ldots$$

and constants

$$g = 0.781789 \ldots, \quad h = 0.113917 \ldots$$

$$\lim_{n\to\infty} sup \frac{a_n}{c^{n^2}} * n^2 = g + h \sim 0.8957 \ldots$$

$$\lim_{n\to\infty} inf \frac{a_n}{c^{n^2}} * n^2 = g - h \sim 0.6678 \ldots$$

In following graph is array "val" defined as

$$val(n) = \frac{a_n}{c^{n^2}} * n^2$$

```
period = 309.3; Show[ListPlot[val],
 Plot[0.781789 + 0.113917 * Cos[2 * Pi * (n / period)], {n, 1, 1000}, PlotStyle → Red],
 Plot[0.781789 + 0.113917 * Cos[2 * Pi * (n / period + 1 / 5)], {n, 1, 1000}, PlotStyle → Green],
 Plot[0.781789 + 0.113917 * Cos[2 * Pi * (n / period + 2 / 5)], {n, 1, 1000}, PlotStyle → Blue],
 Plot[0.781789 + 0.113917 * Cos[2 * Pi * (n / period + 3 / 5)], {n, 1, 1000}, PlotStyle → Black],
 Plot[0.781789 + 0.113917 * Cos[2 * Pi * (n / period + 4 / 5)], {n, 1, 1000}, PlotStyle → Brown]]
```



Numerical results for first 1000 terms.

References:

[1] OEIS - The On-Line Encyclopedia of Integer Sequences
[2] Kotěšovec V., Interesting asymptotic formulas for binomial sums, website 9.6.2013
[3] Kotěšovec V., Asymptotic of a sums of powers of binomial coefficients * x^k, website 20.9.2012
[4] Kotěšovec V., Asymptotic of generalized Apéry sequences with powers of binomial coefficients, website 4.11.2012

This article was published on website http://web.telecom.cz/vaclav.kotesovec/math.htm, 21.9.2013